

Microsoft Excel 2016

Tabellenzellen benennen



Inhaltsverzeichnis

Einleitung.....	2
Benutzerdefinierten Namen festlegen	2
Namen aus Auswahl erstellen	4
Syntaxregeln für Namen.....	5
Einsatz von benutzerdefinierten Namen.....	6
Die Übersicht behalten.....	9
Konstante Werte und Formeln benennen.....	13
Verschachtelte Namen verwenden	16
Relative Bezugsart verwenden.....	16
Benutzerdefinierter Name mit Bezug auf mehrere Blätter.....	17
Benutzerdefinierte Namen für externe Bezüge	17
Anzeige von benutzerdefinierten Namen.....	18

Abbildungsverzeichnis

Abb. 1: <i>Dialogfeld Neuer Name</i>	3
Abb. 2: <i>Das Namenfeld</i>	3
Abb. 3: <i>Die Ausgangstabelle für die finanzmathematische Berechnung</i>	4
Abb. 4: <i>Dialogfeld Namen aus Auswahl erstellen</i>	5
Abb. 5: <i>Das Ergebnis der finanzmathematischen Berechnung.....</i>	5
Abb. 6: <i>Benutzerdefinierten Namen aus einer Liste auswählen.....</i>	7
Abb. 7: <i>Dialogfeld Namen einfügen, beim Dialogfeld Funktionsargumente</i>	8
Abb. 8: <i>Dialogfeld Namen übernehmen</i>	8
Abb. 9: <i>Dialogfeld Gehe zu</i>	10
Abb. 10: <i>Benutzerdefinierten Namen im Namenfeld auswählen.....</i>	11
Abb. 11: <i>Dialogfeld Namen einfügen, beim Symbol In Formel verwenden.....</i>	11
Abb. 12: <i>Liste mit allen benutzerdefinierten Namen und zugehörigen Tabellenzellen.....</i>	11
Abb. 13: <i>Dialogfeld Namens-Manager</i>	12
Abb. 14: <i>Der Einsatz des benutzerdefinierten Namens Ostersonntag</i>	15
Abb. 15: <i>Der benutzerdefinierte Name wird im Zellbereich angezeigt, Zoomgröße: 30%</i>	18

Einleitung

Wenn Sie Formeln in Ihren Tabellen einsetzen und sich dabei auf den Inhalt von Tabellenzellen beziehen, werden Sie üblicherweise die Zellennamen über die Kombination aus Spaltenbuchstaben und Zeilennummer (also z.B. **F3**) angeben. Dabei können Sie sowohl die relative Schreibweise (z.B. **F3**) oder die absolute Schreibweise (z.B. **\$F\$3**) verwenden (siehe auch Skript **Microsoft Excel 2016 – Relative und absolute Bezüge**). Bei großen und umfangreichen Tabellen, die neben vielen konstanten Daten auch viele Formeln enthalten, geht bei Verwendung der Originalzellennamen schnell die Übersicht verloren. Zu einem späteren Zeitpunkt können Sie dann die Formeln nur noch schwer nachvollziehen. Sinnvoller ist es, wenn Sie den Tabellenzellen, deren Werte in den Formeln verwendet werden, benutzerdefinierte Namen geben und diese Namen in den Formeln einsetzen. Dadurch behalten Sie viel besser den Überblick über die Daten und die Berechnungen. Insbesondere beim Übertrag von Daten von einem Tabellenblatt auf ein anderes sind benutzerdefinierte Namen sehr sinnvoll einsetzbar. Sie können sowohl einzelnen Tabellenzellen, als auch Zellbereichen einen benutzerdefinierten Namen geben. Auch konstante Zahlen (z.B. aus der Physik oder Mathematik) können auf diese Weise definiert und in Formeln eingesetzt werden. Das Skript bezieht sich auf Excel **2016** bzw. Excel für Office 365, gilt aber prinzipiell auch für die älteren Excel-Versionen (ab Version 2007).


Benutzerdefinierten Namen festlegen

Wenn Sie einer Tabellenzelle bzw. einem Zellbereich einen benutzerdefinierten Namen geben wollen, bietet Excel Ihnen mehrere Möglichkeiten an. Zunächst zwei Möglichkeiten:

- Mit Hilfe des Dialogfelds **Neuer Name**
- Mit Hilfe des Namenfelds

Zunächst wird die Vorgehensweise mit dem Dialogfeld gezeigt:

1. Wählen Sie die Tabellenzelle aus oder markieren den Zellbereich, den Sie benennen wollen.
2. Wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Namen definieren** (alternativ: , ).
3. Im Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 3) tragen Sie in das Textfeld **Name** den benutzerdefinierten Namen ein. Welche Regeln Sie dabei beachten müssen, lesen Sie im Kapitel **Syntaxregeln für Namen**, Seite 5.
4. **Optional**: Wählen Sie in der Liste **Bereich**, ob der benutzerdefinierte Name in der gesamten Arbeitsmappe gültig sein soll oder nur für ein bestimmtes Tabellenblatt.
5. **Optional**: Tragen Sie in das Textfeld **Kommentar** noch eine zusätzliche Beschreibung für den benutzerdefinierten Namen ein. Hierbei müssen Sie auf keine besonderen Regeln achten. Einzig die Länge ist auf 255 Zeichen beschränkt.
6. **Optional**: Geben Sie im Textfeld **Bezieht sich auf** die Tabellenzelle oder den Zellbereich an, für den der benutzerdefinierte Name gelten soll. Diese Angabe ist aber eigentlich nicht notwendig, wenn Sie bereits in Schritt 1 die Tabellenzelle bzw. den Zellbereich ausgewählt haben, die/den Sie benennen wollen.

Abb. 1: Dialogfeld **Neuer Name**

Standardmäßig ist ein benutzerdefinierter Name in der gesamten Arbeitsmappe gültig (also für alle Tabellenblätter; auch wenn manche Tabellenblätter erst nach Festlegung des benutzerdefinierten Namens eingefügt werden). In diesem Fall darf ein benutzerdefinierter Name nur einmal definiert werden. Wenn Sie also z.B. auf dem Tabellenblatt **Tabelle1** einer Tabellenzelle den Namen **Anlagebetrag** gegeben haben, dürfen Sie auf keinem anderen Tabellenblatt einer anderen Tabellenzelle den benutzerdefinierten Namen **Anlagebetrag** zuweisen. Wenn Sie allerdings im Dialogfeld **Neuer Name** (siehe Abbildung 1) in der Liste **Bereich** den Eintrag **Tabelle1** gewählt haben, ist der benutzerdefinierte Name (nehmen wir erneut das Beispiel **Anlagebetrag**) nur für dieses Tabellenblatt gültig. Sie dürfen also auf einem anderen Tabellenblatt (z.B. **Tabelle2**) einer Tabellenzelle erneut den benutzerdefinierten Namen **Anlagebetrag** geben (wenn Sie auch hier im Dialogfeld **Neuer Name** in der Liste **Bereich** den Eintrag **Tabelle2** gewählt haben). Diesen Vorgang können Sie für weitere Tabellenblätter wiederholen. Somit könnte der benutzerdefinierte Name **Anlagebetrag** sooft definiert werden, wie es Tabellenblätter in der Arbeitsmappe gibt. Allerdings muss von dieser Vorgehensweise abgeraten werden, da Sie mit Sicherheit sehr schnell den Überblick verlieren, wenn ein benutzerdefinierter Name mehr als einmal vorkommt.

Alternativ können Sie den benutzerdefinierten Namen auch über das **Namenfeld** festlegen. Hier die genaue Vorgehensweise:

1. Wählen Sie die Tabellenzelle aus oder markieren den Zellbereich, den Sie benennen wollen.
2. Klicken Sie in das Namenfeld, das sich links neben der Bearbeitungsleiste befindet (siehe Abbildung 2). **Bitte klicken Sie nicht auf den kleinen grauen Pfeil am rechten Ende des Namenfelds.**
3. Geben Sie den benutzerdefinierten Namen ein (Regeln siehe Kapitel **Syntaxregeln für Namen**, Seite 5).
4. Bestätigen Sie die Eingabe mit der Taste .




Abb. 2: Das Namenfeld

Anmerkung: Im Gegensatz zum Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 3) können Sie beim Namenfeld keinen Kommentar angeben und auch nicht bestimmen, ob der Name nur für ein bestimmtes Tabellenblatt gelten soll.

Namen aus Auswahl erstellen

Es gibt noch eine alternative Methode, Tabellenzellen benutzerdefinierte Namen zuzuweisen. Dabei müssen Sie die benutzerdefinierten Namen gar nicht selbst eingeben, sondern Excel übernimmt die Namen aus Einträgen in bestimmten Tabellenzellen. Das soll anhand eines Beispiels näher erklärt werden. Nehmen wir an, Sie wollen ein Sparkonto bei einer Bank eröffnen und bei Kontoeröffnung einen festen Geldbetrag (sagen wir als Beispiel **1.000 €**) auf das Konto einzahlen. Außerdem wollen Sie monatlich einen konstanten Betrag (als Beispiel nehmen wir **50 €**) auf das Konto einzahlen. Die Bank gibt Ihnen für dieses Konto einen jährlichen Zinssatz von **1,5%**, allerdings nur, wenn Sie in den nächsten **5** Jahren kein Geld von dem Konto abheben werden. Sie wollen nun mit Hilfe der finanzmathematischen Funktion **ZW (Zinswert)** ermitteln, wie viel Geld sich am Ende der 5 Jahre auf Ihrem Konto angesammelt hat. Zunächst werden die Ausgangsdaten in Tabellenzellen auf dem Tabellenblatt eingetragen. Links neben jeder dieser Tabellenzellen tragen Sie jeweils eine Bezeichnung ein, mit der der entsprechende Zahlenwert genau bezeichnet wird. Die Tabelle¹ könnte folgendes Aussehen haben (siehe Abbildung 3):

	A	B	C
1			
2	Zinssatz:	1,50%	
3	Zeitraum:	5 Jahre	
4	Monatlicher Dauerauftrag:	-50,00 €	
5	Einmalige Einzahlung:	-1.000,00 €	

Abb. 3: Die Ausgangstabelle für die finanzmathematische Berechnung

Die Zahlenwerte für die finanzmathematische Funktion stehen in diesem Beispiel in den Tabellenzellen **B2** bis **B5**. Diese Zellennamen könnten Sie jetzt ohne Probleme in der Formel verwenden. Es ist aber sinnvoller, wenn Sie jeder dieser Tabellenzellen einen benutzerdefinierten Namen zuweisen und diese benutzerdefinierten Namen in der Formel einsetzen. Dadurch wird die Formel transparenter und leichter verständlicher. Jetzt überlegen Sie vielleicht, wie sollen die 4 Tabellenzellen benannt werden und kommen auf die Idee, die Tabellenzelle **B2** **Zinssatz** zu nennen, der Tabellenzelle **B3** den Namen **Zeitraum** zu geben, die Tabellenzelle **B4** soll den Namen **Monatlicher Dauerauftrag** bekommen und die Tabellenzelle **B5** soll **Einmalige Einzahlung** heißen. Wenn Sie sich die Tabelle in Abbildung 3 genau anschauen, dann stehen diese vier Namen ja bereits in den Tabellenzellen **A2** bis **A5**. Jetzt überlegen Sie vielleicht, kann ich diese Namen in der Spalte **A** für meine Namensdefinition in der Spalte **B** verwenden? Die Antwort ist: ja. Dazu markieren Sie den Zellbereich mit den Tabellenzellen, wo die Namen eingetragen sind und die Tabellenzellen mit den Zahlenwerten (also die Tabellenzellen, die benannt werden sollen). Das wäre im vorliegenden Beispiel der Zellbereich **A2:B5**. Dann wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Aus Auswahl erstellen** (alternativ:   ). Im Dialogfeld **Namen aus Auswahl erstellen** (siehe Abbildung 4, Seite 5) können Sie jetzt bestimmen, wo die Namen im markierten Zellbereich



¹ Bei den beiden Geldbeträgen in **B4** und **B5** steht jeweils ein Minuszeichen vor der Zahl. Das bedeutet zunächst, dass Sie (sinnbildlich) Geld an die Bank abgeben. Das Ergebnis der finanzmathematischen Funktion ist dann ein positiver Zahlenwert (sinnbildlich bekommen Sie das Geld von der Bank ausgezahlt).

stehen, die als benutzerdefinierte Namen verwendet werden sollen. In diesem Beispiel stehen die Namen *links* von den Tabellenzellen mit den Zahlenwerten. Sie müssen also in diesem Fall nichts verändern (Kontrollkästchen **Linker Spalte** ist bereits aktiviert) und können das Dialogfeld direkt bestätigen. Die Tabellenzellen **B2** bis **B5** bekommen nun die entsprechenden Namen aus den Tabellenzellen **A2** bis **A5** zugewiesen, wobei Excel sich an die Regeln von benutzerdefinierten Zellennamen hält (siehe nachfolgendes Kapitel). Excel passt, falls erforderlich, die Namen selbst an.




Abb. 4: Dialogfeld **Namen aus Auswahl erstellen**

Nun können Sie die benutzerdefinierten Namen in der Formel einsetzen. In diesem Beispiel lautet die Formel (z.B. können Sie die Formel in die Tabellenzelle **B7** eintragen):

=ZW(Zinssatz/12;Zeitraum*12;Monatlicher_Dauerauftrag;Einmalige_Einzahlung)

Beachten Sie bitte, dass der Zinssatz durch **12** geteilt und dass der Zeitraum mit dem Wert **12** multipliziert werden muss, da die Funktion **ZW** die Berechnung immer monatsweise durchführt. Das Ergebnis der Berechnung sehen Sie in Abbildung 5.

	A	B	C
1			
2	Zinssatz:	1,50%	
3	Zeitraum:	5 Jahre	
4	Monatlicher Dauerauftrag:	-50,00 €	
5	Einmalige Einzahlung:	-1.000,00 €	
6			
7		4.191,18 €	

Abb. 5: Das Ergebnis der finanzmathematischen Berechnung

Syntaxregeln für Namen

Wenn Sie einer Tabellenzelle oder einem Zellbereich einen benutzerdefinierten Namen zuweisen wollen, müssen Sie sich an ein paar Syntaxregeln halten, die Excel Ihnen vorgibt:

- Das erste Zeichen des Namens muss ein Buchstabe, ein Unterstrich () oder ein umgekehrter Schrägstrich (****) sein.
- Für alle weiteren Zeichen des Namens können Sie Buchstaben, Ziffern, den Unterstrich (), den Punkt (.) oder den umgekehrten Schrägstrich (****) verwenden.
- Andere Zeichen sind nicht erlaubt. Dazu gehört auch das Leerzeichen (da es in Excel eine ganz spezielle Bedeutung als Schnittmengenoperator hat). Verwenden Sie zum Trennen von Wörtern den Unterstrich (z.B. **Monatlicher_Dauerauftrag**), den Punkt oder den umgekehrten Schrägstrich.

- Sie können im benutzerdefinierten Namen Groß- oder Kleinbuchstaben verwenden, aber Excel macht keinen Unterschied zwischen Groß- und Kleinschreibung. Wenn Sie eine Tabellenzelle z.B. **ZEITRAUM** genannt haben, dürfen Sie eine andere Tabellenzelle derselben Arbeitsmappe nicht **Zeitraum** nennen.
- Die maximale Länge eines benutzerdefinierten Namens beträgt 255 Zeichen.
- Die Buchstaben **C**, **c**, **R** und **r**² dürfen nicht als benutzerdefinierte Namen verwendet werden, da sie alle als Kurzform für das Markieren einer Zeile oder Spalte für die aktuelle markierte Tabellenzelle verwendet werden, wenn Sie diese z.B. im Dialogfeld **Gehe zu**³ im Textfeld **Verweis** einsetzen.
- Benutzerdefinierte Namen dürfen nicht mit Zellbezügen identisch sein. Z.B. ist der Name **Z3S7** oder **F17** nicht als benutzerdefinierter Name erlaubt.
- Benutzerdefinierte Namen sollten nicht mit den Namen von Excel-Funktionen identisch sein, da es sonst schnell zu Verwechslungen kommen kann⁴.

Einsatz von benutzerdefinierten Namen

Wenn Sie Tabellenzellen bzw. Zellbereichen benutzerdefinierte Namen zugewiesen haben, können Sie diese benutzerdefinierten Namen nun in beliebig vielen Formeln einsetzen. Allerdings kann es sein, dass ein benutzerdefinierter Name nur für ein Tabellenblatt gültig ist. Wenn Sie den benutzerdefinierten Namen über das Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 3) eingegeben und dort in der Liste **Bereich** den Namen eines Tabellenblatts ausgewählt haben, dann ist der benutzerdefinierte Name nur auf diesem Tabellenblatt gültig. Verwenden Sie den Namen in einer Formel auf einem anderen Tabellenblatt, bekommen Sie als Ergebnis den Fehlerwert **#NAME?**. Wollen Sie den benutzerdefinierten Namen trotzdem in einer Formel auf einem anderen Tabellenblatt einsetzen, müssen Sie dem benutzerdefinierten Namen den Blattnamen zusammen mit dem Ausrufezeichen voranstellen (Beispiel: wenn der benutzerdefinierte Name **Zinssatz** lautet und für das Tabellenblatt **Tabelle1** festgelegt wurde, können Sie auf einem anderen Tabellenblatt die Angabe **Tabelle1!Zinssatz** in einer Formel vornehmen, wodurch eindeutig geklärt ist, wo der Name **Zinssatz** definiert worden ist).

Normalerweise ist ein benutzerdefinierter Name für die gesamte Arbeitsmappe gültig (**aber wirklich auch nur innerhalb dieser einen Arbeitsmappe**). Sie können den Namen also in jeder beliebigen Formel einsetzen, egal auf welchem Tabellenblatt sich die Formel befindet.

² Der Buchstabe **C** bzw. **c** steht für das englische Wort **Column** (dt. Spalte) und **R** bzw. **r** für **Row** (dt. Zeile).

³ Das Dialogfeld **Gehe zu** bekommen Sie, wenn Sie Register **Start** in der Gruppe **Bearbeiten** das Symbol **Suchen und Auswählen** anklicken und den Befehl **Gehe zu** aufrufen. Alternativ benutzen Sie die Funktionstaste .

⁴ Grundsätzlich ist es aber erlaubt, einer Tabellenzelle oder einem Zellbereich den Namen einer Excel-Funktion zu geben. Excel kann trotzdem den benutzerdefinierten Namen von dem Namen der Excel-Funktion unterscheiden. Der Name einer Excel-Funktion besitzt immer ein Klammerpaar unmittelbar hinter dem Funktionsnamen, während ein benutzerdefinierter Name kein Klammerpaar hinter dem Namen besitzt. Trotzdem sollten Sie keine Funktionsnamen als benutzerdefinierte Namen verwenden, da es die Übersichtlichkeit deutlich einschränkt.

Wenn Sie eine Formel in eine Tabellenzelle (oder in einen Zellbereich bei Matrix-Formeln) über die Tastatur eingeben (und nicht über das Dialogfeld **Funktion einfügen**; siehe Skript **Microsoft Excel 2016 – Funktionen (Beispiele)**, Seite 25) und in der Formel einen benutzerdefinierten Namen verwenden, müssen Sie während der Eingabe nicht auf die Groß-/Kleinschreibung achten. Sie können den benutzerdefinierten Namen komplett in Kleinbuchstaben eingeben. Sobald Sie die ersten Zeichen eingegeben haben, bekommen Sie eine Auswahlliste, wo Sie neben Excel-Funktionen auch benutzerdefinierte Namen angezeigt bekommen (siehe Abbildung 6). Sie können nun den benutzerdefinierten Namen bequem aus der Liste auswählen (z.B. bewegen Sie das Maussymbol auf den benutzerdefinierten Namen und machen einen Doppelklick). Wenn Sie den benutzerdefinierten Namen nur anklicken, bekommen Sie in einem Infokasten auch den Kommentar angezeigt (wenn Sie in Schritt 5, Seite 2, einen Kommentar eingegeben haben).





Abb. 6: Benutzerdefinierten Namen aus einer Liste auswählen

Anmerkung: In der Auswahlliste können Sie eine Excel-Funktion von einem benutzerdefinierten Namen noch durch entsprechende Symbole unterscheiden. Vor einer Excel-Funktion sehen Sie das Symbol und vor dem benutzerdefinierten Namen das Symbol .

Sie können einen benutzerdefinierten Namen bei der Eingabe einer Formel auch aus einer Liste auswählen, anstatt ihn über die Tastatur einzutippen. Wenn Sie in der Formeleingabe an der Stelle sind, wo jetzt ein benutzerdefinierter Name eingegeben werden soll, wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **In Formel verwenden** und wählen aus der angebotenen Liste den Namen aus. Alternativ benutzen Sie die Funktionstaste und wählen den Namen aus dem Dialogfeld **Namen einfügen** (siehe Abbildung 7, Seite 8).



Wenn Sie in einer Formel zum ersten Mal eine neue Funktion einsetzen, werden Sie gegebenenfalls den Funktionsassistenten verwenden (Register **Formeln**, Gruppe **Funktionsbibliothek**, Symbol **Funktion einfügen**). Im Dialogfeld **Funktion einfügen** wählen Sie die Funktion aus und im nachfolgenden Dialogfeld **Funktionsargumente** geben Sie in die Textfelder die entsprechenden Daten für die Funktionsargumente ein. Dabei kann es sich um konstante Daten handeln oder auch um Originalzellennamen. Natürlich können Sie auch benutzerdefinierte Namen angeben. Die Namen können Sie entweder über die Tastatur eintragen, wobei Sie den kompletten Namen eingeben müssen, oder Sie benutzen die Funktionstaste und erhalten das Dialogfeld **Namen einfügen**



(siehe Abbildung 7, Seite 8), wo Sie den benutzerdefinierten Namen bequem aus einer Liste auswählen können.




Abb. 7: Dialogfeld **Namen einfügen**, beim Dialogfeld **Funktionsargumente**

Jetzt kann es in der Praxis auch vorkommen, dass Sie eine Tabelle mit konstanten Daten und Formeln haben, wobei Sie in den Formeln auf Tabellenzellen verweisen, indem Sie die Originalnamen (also z.B. **B4** oder **F12**) verwenden. Zu einem späteren Zeitpunkt entscheiden Sie sich, bestimmten Tabellenzellen benutzerdefinierte Namen zu geben. Natürlich wollen Sie jetzt in den Formeln, wo die Daten dieser Tabellenzellen verwendet werden, die Originalzellennamen durch die benutzerdefinierten Namen ersetzen. Sie könnten natürlich nacheinander die Tabellenzellen mit den Formeln auswählen, in die Zellbearbeitung wechseln (z.B. mit der Funktionstaste **F2**) und mühsam jeden Originalzellennamen durch den benutzerdefinierten Namen ersetzen. Das ist für ein paar wenige Formeln vielleicht noch akzeptabel, aber bei sehr vielen Formeln viel zu aufwendig. Das geht einfacher und schneller. Wählen Sie erneut die Tabellenzelle mit der Formel aus, wo die Originalzellennamen durch die benutzerdefinierten Namen ersetzt werden sollen. Diesmal wechseln Sie aber **nicht** in die Zellbearbeitung. Klicken Sie im Register **Formeln** in der Gruppe **Definierte Namen** neben dem Symbol **Namen definieren** auf den kleinen grauen Pfeil und wählen den Befehl **Namen übernehmen**. Im Dialogfeld **Namen übernehmen** (siehe Abbildung 8) wählen Sie alle benutzerdefinierten Namen aus, die in die Formel übernommen werden sollen und bestätigen das Dialogfeld. Sie müssen dabei nicht jeden Originalzellennamen durch einen benutzerdefinierten Namen ersetzen. Wiederholen Sie den ganzen Vorgang für weitere Tabellenzellen mit Formeln, wo Sie die Originalzellennamen durch benutzerdefinierte Namen ersetzen wollen.




Abb. 8: Dialogfeld **Namen übernehmen**

Im Dialogfeld **Namen übernehmen** können Sie noch zusätzliche Einstellungen über Kontrollkästchen und der Schaltfläche vornehmen. Diese Einstellungen sind aber von untergeordneter Bedeutung und werden daher an dieser Stelle nicht näher beschrieben.

Sehr praktisch ist der Einsatz von benutzerdefinierten Namen, wenn Sie einen Zelleninhalt (z.B. die Summe der Einnahmen am Monatsende) von einem Tabellenblatt auf ein anderes übernehmen wollen. Dies soll wieder an einem kleinen Beispiel genauer gezeigt werden. Angenommen, die Summe der Einnahmen befindet sich auf dem Tabellenblatt mit dem Namen **Januar 2016** und zwar in der Tabellenzelle **B54**. Sie haben dieser Tabellenzelle aber zunächst keinen benutzerdefinierten Namen zugewiesen. Nun soll dieser Ergebniswert in eine andere Tabellenzelle auf einem anderen Tabellenblatt (aber noch innerhalb derselben Arbeitsmappe) übernommen werden (aber nicht durch Kopieren des Zellinhalts, sondern mit Hilfe eines Verweises über eine Formel). Auf dem anderen Tabellenblatt (der Name ist an dieser Stelle ohne Bedeutung) wählen Sie die Tabellenzelle aus, wo der Verweis eingetragen werden soll (z.B. die Tabellenzelle **B2**) und geben folgende Formel ein:

='Januar 2016'!B54

Zugegeben, die Formel ist nicht besonders lang, aber Sie müssen mehrere Dinge beachten:

- Die Formel enthält den Namen des Tabellenblatts, wobei der Name in Hochkommas (') eingeschlossen sein muss (wenn er ein oder mehrere Leerzeichen enthält; wie in dem aktuellen Beispiel).
- Der Name der Tabellenzelle auf dem angegebenen Tabellenblatt.
- Das Ausrufezeichen (!) als Trennzeichen zwischen Blattname und Zellenname.

Wenn Sie nun der Tabellenzelle **B54** auf dem Tabellenblatt **Januar 2014** einen benutzerdefinierten Namen zugewiesen haben (z.B. **Einnahmen.Januar**), können Sie auf dem anderen Tabellenblatt in die Tabellenzelle **B2** (aus dem vorherigen Beispiel) folgende Formel schreiben:

=Einnahmen.Januar

Zwar ist die Anzahl der einzugebenden Zeichen in beiden Formeln in etwa gleich, aber Sie müssen hinter dem Gleichheitszeichen nur den benutzerdefinierten Namen eingeben (und den können sie auch bequem aus der angebotenen Liste (siehe Abbildung 6, Seite 7) auswählen). Sie brauchen keinen Blattnamen, keine Hochkommas und kein Ausrufezeichen einzugeben.

Anmerkung: Wenn Sie einen benutzerdefinierten Namen in einer Formel verwenden, handelt es sich immer um eine absolute Bezugsart (siehe auch Skript **Microsoft Excel 2016 – Relative und absolute Bezüge**).

Die Übersicht behalten

Wenn Sie innerhalb einer Arbeitsmappe auf verschiedenen Tabellenblättern eine ganze Reihe an benutzerdefinierten Namen festgelegt haben, verlieren Sie recht schnell die Übersicht. Wenn Sie dann auf einem Tabellenblatt eine Formel sehen und bei den Argumenten handelt es sich um benutzerdefinierte Namen, kann es passieren, dass Sie nicht mehr genau wissen, wo sich die dazugehörigen Tabellenzellen bzw. Zellbereiche befinden. Insbesondere wenn Sie von jemand anderem eine Excel-

Datei bekommen, wo ebenfalls benutzerdefinierte Namen verwendet werden, ist es eigentlich unmöglich herauszufinden, welche Tabellenzellen bzw. Zellbereiche benannt worden sind und welche nicht. Aber keine Sorge, Excel bietet ihnen mehrere Möglichkeiten an, sich die Tabellenzellen bzw. Zellbereiche zu den benutzerdefinierten Namen anzeigen zu lassen:

- Mit Hilfe des Dialogfelds **Gehe zu**
- Mit Hilfe des Namenfelds
- Liste aller benutzerdefinierter Namen erstellen lassen
- Mit Hilfe des Dialogfelds **Namens-Manager**

Bei allen vier Möglichkeiten spielt es keine Rolle, welches Tabellenblatt aktuell aktiv ist.

Wenn Sie das Dialogfeld **Gehe zu** (siehe Abbildung 9) verwenden wollen, wählen Sie im Register **Start** in der Gruppe **Bearbeiten** das Symbol **Suchen und Auswählen** und den Befehl **Gehe zu**. (alternativ: ). Im Dialogfeld **Gehe zu** wählen Sie den benutzerdefinierten Namen in der Liste **Gehe zu** und bestätigen das Dialogfeld. Excel wechselt automatisch zum entsprechenden Tabellenblatt und setzt den Auswahlrahmen auf die zugehörige Tabellenzelle bzw. markiert den zugehörigen Zellbereich.





Abb. 9: Dialogfeld **Gehe zu**

Alternativ können Sie sich die Liste der benutzerdefinierten Namen auch im Namenfeld anzeigen lassen. Klicken Sie im Namenfeld auf den kleinen grauen Pfeil am rechten Ende des Namenfeld (siehe Abbildung 10, Seite 11) und Sie sehen die Liste mit den benutzerdefinierten Namen. Jetzt müssen Sie nur noch den gewünschten Namen anklicken. Wenn die Liste sehr lange benutzerdefinierte Namen enthält, werden diese Namen meistens nicht komplett angezeigt, da die Breite des Namenfelds relativ klein ist. Bewegen Sie das Mausymbol auf die drei kleinen grauen Punkte (☰) zwischen Namenfeld und Bearbeitungsleiste und vergrößern bzw. verkleinern bei gedrückter linker Maustaste die Breite des Namenfeld (die Breite der Bearbeitungsleiste wird ebenfalls geändert).

Bei der dritten Möglichkeit können Sie sich alle benutzerdefinierten Namen und die zugehörigen Tabellenzellen bzw. Zellbereiche auf einem Tabellenblatt einfügen lassen. Dabei muss es sich nicht zwangsläufig um ein leeres Tabellenblatt handeln. Es müssen nur zwei Spalten auf dem Tabellenblatt frei sein. Wählen Sie im ausgesuchten Tabellenblatt eine freie Tabellenzelle aus, wo Excel mit der Eingabe der benutzerdefinierten Namen beginnen soll. Dann wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **In Formel verwenden** und den Befehl **Namen einfügen** (alt. ). Im Dialogfeld **Namen einfügen** (siehe Abbildung 11, Seite 11) klicken Sie



auf die Schaltfläche **Liste einfügen**. Das Ergebnis sehen Sie in Abbildung 12 (Spaltenbreiten wurden nachträglich angepasst).




Abb. 10: Benutzerdefinierten Namen im Namenfeld auswählen




Abb. 11: Dialogfeld **Namen einfügen**, beim Symbol **In Formel verwenden**

	A	B	C	D	E	F	G
1	Datum	=41266					
2	Einmalige_Einzahlung	=Tabelle5!\$B\$5					
3	Einnahmen.Quartal	=Tabelle1!\$B\$2:\$B\$5					
4	Fachbereiche	={"Physik","Mathematik","Chemie","Psychologie","Jura","Theologie"}					
5	Gesamteinnahmen	=SUMME(Einnahmen)					
6	Monatlicher_Dauerauftrag	=Tabelle5!\$B\$4					
7	Zinssatz	=Tabelle5!\$B\$2					

Abb. 12: Liste mit allen benutzerdefinierten Namen und zugehörigen Tabellenzellen

Bei der vierten Möglichkeit können sie sich alle benutzerdefinierten Namen im Dialogfeld **Namens-Manager** (siehe Abbildung 13, Seite 12) anzeigen lassen. Sie sehen nicht nur die benutzerdefinierten Namen und die zugehörigen Tabellenzellen bzw. Zellbereiche, sondern auch welchen Inhalt diese Tabellenzellen bzw. Zellbereiche enthalten und auch die Kommentare (sofern bei der Erstellung des benutzerdefinierten Namens ein Kommentar in Schritt 5, Seite 2, angegeben wurde). Wählen Sie im Register **Formeln** in der Gruppe **Definierte Namen** das Symbol **Namens-Manager**.







Abb. 13: Dialogfeld **Namens-Manager**

Die Tabellendarstellung im Dialogfeld **Namens-Manager** hat folgenden Aufbau:

Spalte	Bedeutung
Name	In dieser Spalte werden die benutzerdefinierten Namen aufgelistet.
Wert	Hier sehen Sie die aktuellen Werte, die in den Tabellenzellen bzw. Zellbereichen stehen, die zu den benutzerdefinierten Namen gehören. Wenn es sich um einen Zellbereich mit mehreren Werten handelt, werden die Werte in Matrix-Schreibweise mit Punkten bzw. Semikola getrennt und in geschweiften Klammern eingeschlossen dargestellt (siehe auch Skript Microsoft Excel 2016 – Matrix-Funktionen).
Bezieht sich auf	In dieser Spalte sehen Sie die zugehörigen Tabellenzellen bzw. Zellbereiche und die Blatt-namen, wo sich die Tabellenzellen bzw. Zellbereiche befinden.
Bereich	In dieser Spalte können Sie sehen, ob der benutzerdefinierte Name global für die gesamte Arbeitsmappe gültig ist oder nur für eine bestimmtes Tabellenblatt.
Kommentar	Wurde in Schritt 5, Seite 2, ein Kommentar für den benutzerdefinierten Namen festgelegt, wird dieser in der letzten Spalte des Namens-Managers angezeigt.

Sie können bei Bedarf die Spaltenbreiten im Namens-Manager anpassen. Bewegen Sie das Maussymbol in der Überschriftzeile (da wo **Name**, **Wert**, **Bezieht sich auf**, **Bereich** und **Kommentar** steht) auf die Trennlinie zwischen zwei Spalten und verändern mit der Maus die Breite der Spalte bei gedrückter linker Maustaste. Wenn Sie das Maussymbol auf die Trennlinie bewegen und einen Doppelklick ausführen, wird die optimale Spaltenbreite eingestellt. Damit die Spaltenbreitenänderung auch einen Sinn ergibt, können Sie auch die Breite (und bei Bedarf auch die Höhe) des Dialogfelds über den Rahmen des Dialogfelds mit der Maus anpassen.

Der Namens-Manager bietet noch eine Reihe weiterer Möglichkeiten. Klicken Sie auf die Schaltfläche **Neu...** und Sie können einen neuen benutzerdefinierten Namen anlegen. Die Vorgehensweise ist dabei identisch mit den Schritten auf Seite 2. Wenn Sie einen benutzerdefinierten Namen ändern wollen (oder die zugehörige Tabellenzelle bzw. der zugehörige Zellbereich), wählen Sie zunächst den benutzerdefinierten Namen in der Auflistung aus und klicken auf die Schaltfläche **Bearbeiten...** und erhalten das Dialogfeld **Namen bearbeiten** (Aufbau und Nutzung ist identisch mit dem Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 3)). Sie können natürlich auch jederzeit einen benutzerdefinierten Namen löschen. Wählen Sie den benutzerdefinierten Namen in der Auflistung und klicken

auf die Schaltfläche . Excel fragt Sie sicherheitshalber nochmal, ob Sie den benutzerdefinierten Namen auch wirklich löschen wollen. Beim Löschen eines benutzerdefinierten Namens sollten Sie etwas vorsichtig sein. Wird der benutzerdefinierte Name in einer oder mehreren Formeln eingesetzt, erhalten Sie bei diesen Formeln als Ergebnis den Fehlerwert **#NAME?**. Wenn Sie also nicht sicher sind, lassen Sie den benutzerdefinierten Namen lieber im Namens-Manager.

Wenn die Arbeitsmappe sehr viele benutzerdefinierte Namen enthält, kann die Auflistung der Namen im Namens-Manager sehr schnell unübersichtlich werden. Deshalb können Sie die Auflistung filtern. Klicken Sie in der rechten oberen Ecke des Dialogfelds auf die Schaltfläche  und Sie erhalten eine Befehlsliste. Die Befehle haben folgende Bedeutungen:

Befehl	Bedeutung
Filter löschen	Die Filterung wird entfernt. Es werden wieder alle benutzerdefinierte Namen im Dialogfeld aufgelistet.
Auf Arbeitsblatt ausgelegte Namen	Es werden nur die benutzerdefinierten Namen aufgelistet, die für ein bestimmtes Tabellenblatt eingeschränkt nutzbar sind.
Auf Arbeitsmappe ausgelegte Namen	Es werden nur die benutzerdefinierten Namen aufgelistet, die in der gesamten Arbeitsmappe benutzt werden können.
Namen mit Fehlern	Es werden nur die benutzerdefinierten Namen aufgelistet, die auf Tabellenzellen bzw. Zellbereiche verweisen, die Fehlerwerte enthalten.
Namen ohne Fehler	Es werden nur die benutzerdefinierten Namen aufgelistet, die auf Tabellenzellen bzw. Zellbereiche verweisen, die keine Fehlerwerte enthalten.
Definierte Namen	Es werden nur die benutzerdefinierten Namen von Tabellenzellen bzw. Zellbereichen aufgelistet.
Tabellennamen	Es werden nur die benutzerdefinierten Namen von Tabellen aufgelistet (siehe Skript Microsoft Excel 2016 – Tabellen). Nicht zu verwechseln mit Blattnamen.

Sie können die Auflistung im Namens-Manager auch sortieren. Dabei können Sie sich auch aussuchen, nach welcher Spalte (**Name**, **Wert**, **Bezieht sich auf**, usw.) die Sortierung erfolgen soll. Klicken Sie einmal auf den Überschriftenamen der Spalte, die Sie sortiert haben wollen. Dabei werden die Angaben in der betreffenden Spalte auf- oder absteigend sortiert. Klicken Sie nochmals auf die Spaltenüberschrift, wechselt die Sortierreihenfolge.

Konstante Werte und Formeln benennen

Neben Tabellenzellen und Zellbereichen können Sie auch konstante Werte (Zahl, Text, Datum, Uhrzeit, usw.) mit einem benutzerdefinierten Namen definieren. Beispielsweise können Sie physikalische Konstanten definieren, die Sie dann in physikalischen Berechnungen einsetzen können. Am Beispiel der Konstante für die Lichtgeschwindigkeit soll die Vorgehensweise beim Definieren von Konstanten näher gezeigt werden.

- Benutzen Sie z.B. die Tastenkombination    um das Dialogfeld **Neuer Name** (siehe Abbildung 1, Seite 3) aufzurufen.

2. Tragen Sie in das Textfeld **Name** den Namen für die Konstante ein (in diesem Beispiel wird der Name **Lichtgeschwindigkeit** genommen⁵).
3. Löschen Sie den Inhalt im Textfeld **Bezieht sich auf** und geben den konstanten Wert (in diesem Beispiel **299792,458**) ein und bestätigen das Dialogfeld.

Jetzt können Sie den Namen **Lichtgeschwindigkeit** wie jeden anderen benutzerdefinierten Namen in beliebigen Formeln einsetzen.

Anmerkung: Benutzerdefinierte Namen gelten nur innerhalb der Arbeitsmappe. Das gilt auch bei Konstanten. Wenn Sie bestimmte Konstante global in sehr vielen Arbeitsmappen einsetzen wollen, müssen Sie dafür benutzerdefinierte Funktionen mit der Programmiersprache **VBA** (Visual Basic for Applications; siehe Skript **Microsoft Excel 2016 – Benutzerdefinierte Funktionen**) erstellen.

Sie können auch komplette Listen als konstante Daten bereitstellen. Die Listeneinträge können Sie später z.B. als Spaltenüberschriften für Tabellen verwenden. Als Beispiel soll eine Liste mit Adressfeldern für Personendaten (also *Name, Vorname, Straße, PLZ, Ort*, usw.; wie viele Einträge es genau gibt, können Sie selbst festlegen) angelegt werden. Analog zum Beispiel mit der Lichtgeschwindigkeit öffnen Sie das Dialogfeld **Neuer Name**, geben in das Textfeld **Name** einen benutzerdefinierten Namen ein (z.B. **Adresszeile**) und geben in das Textfeld **Bezieht sich auf** folgenden Eintrag ein:

={"Name"."Vorname"."Straße"."PLZ"."Ort".Tel-Nr"."E-Mail-Adresse"}

Wie Sie an dem Beispiel sehen können, werden die Listenelemente in ein geschweiftes Klammerpaar (**{ }**) eingeschlossen. Jedes Listenelement ist zudem in doppelte Anführungszeichen (**"**) eingeschlossen. Die Listenelemente sind durch Punkte (**.**) getrennt. Siehe auch Skript **Microsoft Excel 2016 – Matrix-Funktionen**.

Wenn Sie jetzt diese Listenelemente in einem (neuen) Tabellenblatt brauchen, markieren Sie dort zunächst einen Zellbereich mit sieben nebeneinanderliegenden Tabellenzellen (z.B. **A1:G1**; es sind deshalb sieben Tabellenzellen, da die vorgegebene Beispielliste sieben Elemente enthält). Dann geben Sie die Formel

=Adresszeile

ein und bestätigen die Eingabe mit der Tastenkombination . Es handelt sich hierbei im Grunde um eine Matrix-Funktion (siehe Skript **Microsoft Excel 2016 – Matrix-Funktionen**).

Es kann eventuell vorkommen, dass Sie die Elemente dieser Liste auch mal in Tabellenzellen in einer Spalte (also untereinander) eingeben möchten. In diesem Fall markieren Sie sieben untereinanderliegende Tabellenzellen und geben folgende Formel ein:

=MTRANS(Adresszeile)

⁵ Wenn Sie jetzt daran denken, dass es für die meisten physikalischen Konstanten eine Abkürzung gibt (die Lichtgeschwindigkeit wird mit dem Buchstaben *c* gekennzeichnet) und Sie wollen diese Abkürzung als benutzerdefinierten Name verwenden, dann muss daran erinnert werden, dass die Buchstaben **C, c, R** und **r** nicht für benutzerdefinierte Namen genommen werden dürfen.

Die Eingabe schließen Sie ebenfalls mit der Tastenkombination  ab. Mit Hilfe der Funktion **MTRANS** werden die Listenelemente aus der Zeilenanordnung in die Spaltenanordnung transponiert. Würden Sie in diesem Beispiel die Funktion **MTRANS** weglassen, sehen Sie in allen sieben Tabellenzellen denselben Eintrag: **Name**.

Anmerkung: Wenn Sie konstante Listen erstellen wollen, wobei die Listenelemente später immer in einer Spalte untereinanderstehen sollen, müssen Sie im Textfeld **Bezieht sich auf** (Dialogfeld **Neuer Name**) zwischen den Listeneinträgen anstelle der Punkte das Semikolon (;) verwenden.

Sie können im Dialogfeld **Neuer Name** im Textfeld **Bezieht sich auf** hinter dem Gleichheitszeichen eine Formel eingeben. Die Formel kann auch Zellennamen (z.B. **B3**) bzw. benutzerdefinierte Zellennamen enthalten. Wird der Name dann in einer Formel verwendet (die sich in einer Tabellenzelle befindet), wird zunächst die zum Namen gehörende Formel ausgewertet und das Ergebnis dann in der Formel weiterverwendet, die in der Tabellenzelle steht. Dazu erneut ein kleines Beispiel:

Excel enthält zwar eine ganze Reihe an Datumsfunktionen, aber mit keiner ist es möglich, das Datum für den Ostersonntag zu berechnen. Dafür müssen Sie eine verhältnismäßig komplexe Formel erstellen (unter Zuhilfenahme von verschiedenen Excel-Funktionen). Die Formel lautet⁶ (in diesem Beispiel wird vorausgesetzt, dass sich die Jahreszahl in der Tabellenzelle **B1** befindet):

=KÜRZEN(DATUM(A1;3;56-REST(REST(A1;19)*10,63+5;29))/7)*7+1

Diese Funktion jedes Mal einzugeben, wenn sie in der Arbeitsmappe gebraucht wird, ist zu aufwendig, zumal hierbei auch sehr leicht Eingabefehler passieren können. Deshalb wird die Formel im Dialogfeld **Neuer Name** eingegeben und als Name bietet sich **Ostersonntag** an. Nun kann der Name überall auf dem Tabellenblatt eingesetzt werden. Der Name kann auch in anderen Formeln eingesetzt werden (z.B. zum Berechnen der Feiertage *Karfreitag*, *Ostermontag*, *Christi Himmelfahrt*, *Pfingsten* und *Fronleichnam*; diese Feiertage ergeben sich durch Subtraktion bzw. Addition von genau vorgegebenen Anzahl von Tagen). Abbildung 14 zeigt die Feiertage für das Jahr 2016.

	A	B	C	D	E	F	G
1	Jahr:	2016		Karfreitag:	Freitag, 25. März 2016		Formel in E1: =Ostersonntag-2
2				Ostersonntag:	Sonntag, 27. März 2016		Formel in E2: =Ostersonntag
3				Ostermontag:	Montag, 28. März 2016		Formel in E3: =Ostersonntag+1
4				Christi Himmelfahrt:	Donnerstag, 5. Mai 2016		Formel in E4: =Ostersonntag+39
5				Pfingstsonntag:	Sonntag, 15. Mai 2016		Formel in E5: =Ostersonntag+49
6				Pfingstmontag:	Montag, 16. Mai 2016		Formel in E6: =Ostersonntag+50
7				Fronleichnam:	Donnerstag, 26. Mai 2016		Formel in E7: =Ostersonntag+60

Abb. 14: Der Einsatz des benutzerdefinierten Namens **Ostersonntag**

Anmerkung: Die Formel für die Berechnung des Ostersonntags liefert kein wirkliches Datum, sondern nur eine Zahl. Diese muss dann noch mit einem Zahlenformat als Datum dargestellt werden. In Abbildung 14 wurde das Zahlenformat **TTTT, T. MMMM JJJJ** für den Zellbereich **E1:E7** verwendet.

⁶ Die Formel funktioniert nur, wenn im Dialogfeld **Excel-Optionen** in der Kategorie **Erweitert** in der Gruppe **Beim Berechnen dieser Arbeitsmappe** das Kontrollkästchen **1904-Datumswerte verwenden** **deaktiviert** ist.

Sie können jetzt in der Tabellenzelle **B1** die Jahreszahl ändern (bis maximal 2099) und sehen dann im Zellbereich **E1:E7** die Datumsangaben für die entsprechenden Feiertage. Bedenken Sie aber bitte immer, dass der benutzerdefinierte Name nur in der aktuellen Arbeitsmappe gültig ist. Wenn Sie die Formel für die Berechnung des Osterdatums global in beliebigen Arbeitsmappen einsetzen wollen, müssen Sie eine benutzerdefinierte Funktion erstellen (siehe Skript **Microsoft Excel 2016 – Benutzerdefinierte Funktionen**, Seite 5).

Verschachtelte Namen verwenden

Excel bietet Ihnen die Möglichkeit an, einen benutzerdefinierten Namen in Bezug auf einen anderen benutzerdefinierten Namen zu verwenden. Das klingt etwas verwirrend, daher hier wieder ein konkretes Beispiel. Angenommen Sie haben in einem beliebigen Tabellenblatt im Zellbereich **B2:B13** die Einnahmen für die Monate *Januar* bis *Dezember* eines Jahres (stehen z.B. im Zellbereich **A2:A13**) eingetragen (in der Tabellenzelle **B1** steht *Einnahmen* als Spaltenüberschrift). Nun soll die Summe der Einnahmen ermittelt werden. Das Ergebnis der Berechnung brauchen Sie aber jetzt mehrmals und evtl. auch in verschiedenen Tabellenblättern der Arbeitsmappe. Zunächst markieren Sie den Zellbereich **B2:B13** und geben ihm einen benutzerdefinierten Namen (z.B. **Einnahmen**). Jetzt könnten Sie in jeder beliebigen Tabellenzelle (unabhängig vom Tabellenblatt) die Formel

=SUMME(Einnahmen)

eintragen. Anstelle der Formel können Sie aber auch einen weiteren benutzerdefinierten Namen verwenden. Öffnen Sie dazu das Dialogfeld **Neuer Name**, tragen in das Textfeld **Name** einen benutzerdefinierten Namen ein (z.B. **Gesamteinnahmen**) und geben in das Textfeld **Bezieht sich auf** die Formel **=SUMME(Einnahmen)** ein und bestätigen das Dialogfeld. Wenn Sie jetzt die Summe der Einnahmen in verschiedenen Tabellenzellen benötigen, geben Sie in diese Tabellenzellen die Formel

=Gesamteinnahmen

ein und erhalten die *Summe der Einnahmen*. Zugegeben, die Ersparnis bei der Formeleingabe ist in diesem Beispiel nicht besonders hoch. Wenn es sich aber um eine komplexere Berechnung (z.B. mit mehreren ineinander verschachtelten Funktionen) handelt, müssten Sie die Formel also nur einmal eingeben (im Textfeld **Bezieht sich auf**) und nicht mehrere Male in verschiedenen Tabellenzellen der Arbeitsmappe.

Relative Bezugsart verwenden

In der Anmerkung auf Seite 9 können Sie nachlesen, dass, wenn Sie einer Tabellenzelle einen benutzerdefinierten Namen zuweisen, es sich dabei immer um einen absoluten Bezug handelt. Wenn Sie diesen benutzerdefinierten Namen in irgendwelchen Formeln verwenden, ist immer der Bezug auf diese eine ganz bestimmte Tabellenzelle festgelegt. Excel bietet aber auch die Möglichkeit an, relative Bezüge (oder auch Mischformen) bei der Verwendung von benutzerdefinierten Namen einzusetzen. Auch in diesem Fall wird für eine bessere Verständlichkeit ein konkretes Beispiel genommen.

In einer Tabelle sind in Spalte **A** (ab Tabellenzelle **A2**) die Einnahmen und in Spalte **B** (ab Tabellenzelle **B2**) die Ausgaben eingetragen. In Spalte **C** (genauer: ab Tabellenzelle **C2**) soll nun jeweils der Differenzbetrag zwischen Einnahmen und Ausgaben (genauer **=B2-A2**, **=B3-A3**, **=B4-A4**, usw.) ermittelt werden. Dabei soll ein benutzerdefinierter Name eingesetzt werden. Öffnen Sie das Dialogfeld **Neuer Name**, geben in das Textfeld **Name** einen beliebigen benutzerdefinierten Namen ein (z.B. **Differenzbetrag**) und in das Textfeld **Bezieht sich auf** schreiben Sie die Formel **=B2-A2** (beachten Sie bitte, dass nur vor den Spaltenbuchstaben ein Dollarzeichen angegeben ist; die Zeilennummer ist also relativ) und bestätigen das Dialogfeld. Nun wählen Sie die Tabellenzelle **C3** aus und geben als Formel

=Differenzbetrag

ein. Sie können anschließend den Eintrag von Tabellenzelle **C2** mit Hilfe des *automatischen Ausfüllens* (siehe Skript **Microsoft Excel 2016 – Automatisches Ausfüllen**) in die Tabellenzellen **C3**, **C4**, **C5**, usw. kopieren.

Benutzerdefinierter Name mit Bezug auf mehrere Blätter

Excel erlaubt es Ihnen auch, benutzerdefinierte Namen mit Bezug auf mehrere Blätter festzulegen. Auch hierfür direkt ein konkretes Beispiel. Angenommen Sie haben verschiedene Messreihen durchgeführt und die Ergebnisse in verschiedenen Tabellenblättern eingetragen. Dabei haben Sie aber auf jedem Tabellenblatt immer denselben Zellbereich verwendet (**das ist wichtig!**). Beispielsweise stehen die Messwerte auf allen Tabellenblättern im Zellbereich **A1:B10**. Als Beispiel wird angenommen, es gibt drei Messreihen, also drei Tabellenblätter. Diese haben die Blattnamen *Messreihe1*, *Messreihe2* und *Messreihe3* (alle Blattnamen ohne Leerzeichen). Sie möchten gerne den Mittelwert aus allen Messdaten aller Messreihen ermitteln. Zunächst wird der benutzerdefinierte Name festgelegt. Im Dialogfeld **Neuer Name** tragen Sie in das Textfeld **Name** einen beliebigen benutzerdefinierten Namen ein (z.B. **Messreihen**) und in das Textfeld **Bezieht sich auf** **=Messreihe1:Messreihe3!\$A\$1:\$B\$10** und bestätigen das Dialogfeld. Auf einem weiteren Tabellenblatt (Blattname ist ohne Bedeutung) wählen Sie die Tabellenzelle aus, wo der Mittelwert der Messdaten eingetragen werden soll und geben folgende Formel ein:

=MITTELWERT(Messreihen)

Benutzerdefinierte Namen für externe Bezüge

Sie können auch Bezüge auf Tabellenzellen oder Zellbereiche festlegen, die sich in anderen Arbeitsmappen befinden. Damit der Bezug auch eindeutig ist, muss neben dem Zellennamen auch der Blattname und der Dateiname der Arbeitsmappe angegeben werden. Dabei müssen Sie noch unterscheiden, ob die Arbeitsmappe, für die der Bezug gelten soll, in Excel geöffnet ist oder nicht. Nehmen wir einmal an, Sie wollen in der aktuellen Arbeitsmappe einen Bezug für die Tabellenzelle **C5** (besitzt den benutzerdefinierten Namen **Gesamtergebnis**) herstellen, die sich auf dem Tabellenblatt **Jahresabschluss** in der Arbeitsmappe **Geschäftsjahr 2016.xlsx** befindet. Die Arbeitsmappe ist auf dem Festplattenlaufwerk **D:** im Ordner **Geschäftsdaten** abgelegt. Die Arbeitsmappe **Geschäftsjahr 2016.xlsx** ist aktuell geöffnet. In der aktuellen Arbeitsmappe wählen Sie nun eine Tabellenzelle auf einem beliebigen Tabellenblatt aus und geben folgende Formel ein:

=[Geschäftsjahr 2016.xlsx]Jahresabschluss!C5

Alternativ können Sie auch folgenden Bezug eingeben:

=[Geschäftsjahr 2016.xlsx]Jahresabschluss!Gesamtergebnis

Es wird ja in diesem Beispiel davon ausgegangen, dass die Tabellenzelle **C5** den benutzerdefinierten Namen **Gesamtergebnis** besitzt. Wenn die Arbeitsmappe **Geschäftsjahr 2013.xlsx** **nicht** geöffnet ist, müssen Sie folgenden Bezug verwenden:

=D:\Geschäftsdaten\[Geschäftsjahr 2016.xlsx]Jahresabschluss!C5 oder

=D:\Geschäftsdaten\[Geschäftsjahr 2016.xlsx]Jahresabschluss!Gesamtergebnis

Anzeige von benutzerdefinierten Namen

Zum Abschluss dieses Themas noch eine kleine Besonderheit bei der Anzeige von benutzerdefinierten Namen. Diese Besonderheit gilt aber nur für Zellbereiche, denen Sie benutzerdefinierte Namen zugewiesen haben. Wenn Sie die Zoomgröße auf **30%** oder kleiner einstellen, wird in dem Zellbereich, dem Sie den benutzerdefinierten Namen zugewiesen haben, der Name sichtbar angezeigt. Wird die Zoomgröße auf einen Wert größer als **30%** eingestellt, ist der benutzerdefinierte Name nicht mehr zu sehen. In Abbildung 15 wurde einem umfangreichen Zellbereich ein benutzerdefinierter Name (in diesem Beispiel der Name **Zellbereich**) zugewiesen und anschließend die Zoomgröße auf **30%** eingestellt. Da bei einer so kleinen Zoomgröße der Zellinhalt eh nicht mehr lesbar ist, wurde auf einen Inhalt der Tabellenzellen verzichtet. Um den Zellbereich besser erkennen zu können, wurde hierfür die Füllfarbe Gelb genommen.



Abb. 15: Der benutzerdefinierte Name wird im Zellbereich angezeigt, Zoomgröße: 30%